


DRAUGHTS



CLASSIFICATION: Simulation Game

A two player game of draughts for all ages, suited for the novice and experienced alike. The game is played using a mock board and the player moves his player to the appropriate position.

The play is controlled with the following keys:

- 'A' to move the cursor up
- 'Z' to move the cursor down
- ',' to move the cursor left
- ',' to move the cursor right

Use the 'SPACE' bar to deposit or pick up the player and the 'C' to change a move. Press 'SHIFT' and 'Q' to replay or quit.

PROGRAMMING SUGGESTIONS

The game could of course be changed in many ways. As it stands it plays a reasonable game. It could be altered to play against the computer, have different levels of play and have time differing time limits for moving in each level of play.

PROGRAM

Variables

A, B	Cursor coordinates
C	Colour
REPLAY	Flag for replay
FLAG	Has 'C' been pressed?
X(8,8), Y(8,)	Coordinates of each square
COL (8,8)	Paper colours for each square
CONTENTS(8,8)	Contents of each square
MX(3000), MY(3000)	Moves coordinates
COUNT	Moves count
BPIECES	Number of black pieces
RPIECES	Number of red pieces taken
N	Temp
PIECES\$	String for draughts piece
SP\$	Space string
T	Temp
FROMX, FROMY	Coordinates of location to move piece from
DESX, DESY	Coordinates to move piece to
BLOCK\$	String to draw

Program Structure

Lines	Function/Activity
30 — 70	Call initialise
100 — 490	Main loop
520 — 660	Cursor control
690 — 770	Replay
800 — 900	Draw board
930 — 980	Block routine
1010 — 1130	Initial initialisation
1160 — 1440	Replay initialisation
1470 — 1520	Menu

DRAUGHTS

Call initialise Game

```

40   GOSUB 1010
50   A = 1 : B = 1
60   COUNT = 1
    
```

Main program loop

```

70   :
100  GOSUB 120
110  GOTO 100
120  C = 4
130  LOCATE 1, 23
140  PEN C : PAPER 0 : PRINT "RED'S MOVE▲▲"
150  IF REPLAY <> 1 THEN GOSUB 520 : PRINT CHR$(7) ELSE GOTO
    180
160  IF FLAG = 1 THEN GOTO 120
170  IF CONTENTS(A, B) <> C THEN GOTO 150 ELSE FROMX = A : F
    ROMY = B
180  IF REPLAY <> 1 THEN GOSUB 520 : PRINT CHR$(7) ELSE GOSU
    B 690
190  IF FLAG = 1 THEN GOTO 120
200  IF COL(A, B) = 3 OR CONTENTS(A, B) <> 0 OR A > FROMX +
    2 OR A < FROMX - 2 OR A = FROMX THEN GOTO 180
210  DESX = A : DESY = B
220  IF FROMX < B THEN IF (A = FROMX + 2 AND CONTENTS(FROMX
    + 1, FROMY + ((DESY - FROMY) / 2)) <> 5) THEN GOTO 180
230  IF FROMX > 1 THEN IF (A = FROMX - 2 AND CONTENTS(FROMX
    - 1, FROMY + ((DESY - FROMY) / 2)) <> 5) THEN GOTO 180
240  PEN C : CONTENTS(A, B) = C : PAPER COL(A, B) : BLOCK$ =
    PIECE$ : GOSUB 930
250  CONTENTS(FROMX, FROMY) = 0 : A = FROMX : B = FROMY : PA
    PER COL(A, B) : BLOCK$ = SP$ : GOSUB 930
260  IF REPLAY <> 1 THEN MX(COUNT) = FROMX : MY(COUNT) = FRO
    MY : COUNT = COUNT + 1 : MX(COUNT) = DESX : MY(COUNT) =
    DESY : COUNT = COUNT + 1
270  IF ABS(FROMY - DESY) = 2 THEN CONTENTS(FROMX + ((DESX -
    FROMX) / 2), (FROMY + ((DESY - FROMY) / 2))) = 0 : BPIE
    CES = BPIECES + 1 : LOCATE BPIECES, 21 : PEN 5 : PAPER 0
    : PRINT CHR$(231) ELSE 290
280  A = FROMX + ((DESX - FROMX) / 2) : B = FROMY + ((DESY
    - FROMY) / 2) : PAPER COL(A, B) : BLOCK$ = SP$ : GOSUB
    930
290  IF BPIECES = 12 THEN LOCATE 1, 23 : PAPER 0 : PRINT "RE
    D WINS!!!▲▲▲▲" : FOR N = 1 TO 10 : FOR T = 1 TO 1000 STE
    P 50 : SOUND 129, T, 10, 15 : NEXT T : NEXT N : FOR T =
    1 TO 3000 : NEXT T : GOTO 1470
300  C = 5
310  LOCATE 1, 23
320  PEN C : PAPER 0 : PRINT "BLUE'S MOVE▲▲"
330  IF REPLAY <> 1 THEN GOSUB 520 : PRINT CHR$(7) ELSE 360
340  IF FLAG = 1 THEN GOTO 300
350  IF CONTENTS(A, B) <> C THEN GOTO 330 ELSE FROMX = A : F
    ROMY = B
360  IF REPLAY <> 1 THEN GOSUB 520 : PRINT CHR$(7) ELSE GOSU
    B 690
    
```

```

370 IF FLAG = 1 THEN GOTO 300
380 IF COL(A, B) = 3 OR CONTENTS(A, B) (<> 0 OR A > FROMX +
2 OR A < FROMX - 2 OR A = FROMX THEN GOTO 360
390 DESX = A : DESY = B
400 IF FROMX < 8 THEN IF (A = FROMX + 2 AND CONTENTS(FROMX
+ 1, FROMY + ((DESY - FROMY) / 2)) (<> 4) THEN GOTO 360
410 IF FROMX > 1 THEN IF (A = FROMX - 2 AND CONTENTS(FROMX
- 1, FROMY + ((DESY - FROMY) / 2)) (<> 4) THEN GOTO 360
420 PEN C : CONTENTS(A, B) = C : PAPER COL(DESX, DESY) : A
= DESX : B = DESY : BLOCK$ = PIECE$ : GOSUB 930
430 IF REPLAY (<> 1 THEN MX(COUNT) = FROMX : MY(COUNT) = FRO
MY : COUNT = COUNT + 1 : MX(COUNT) = DESX : MY(COUNT) =
DESY : COUNT = COUNT + 1
440 CONTENTS(FROMX, FROMY) = 0 : A = FROMX : B = FROMY : PA
PER COL(A, B) : BLOCK$ = SP$ : GOSUB 930
450 IF ABS(FROMY - DESY) = 2 THEN CONTENTS(FROMX + ((DESX -
FROMX) / 2), (FROMY + ((DESY - FROMY) / 2))) = 0 : RPIE
CES = RPIECES + 1 : LOCATE 20, RPIECES : PEN 4 : PAPER 0
: PRINT CHR$(231) ELSE 470
460 A = FROMX + ((DESX - FROMX) / 2) : B = FROMY + ((DESY
- FROMY) / 2) : PAPER COL(A, B) : BLOCK$ = SP$ : GOSUB
930
470 IF RPIECES = 12 THEN LOCATE 1, 23 : PAPER 0 : PRINT "BL
UE WINS!! ▲▲▲▲" : FOR N = 1 TO 10 : FOR T = 1 TO 1000 STE
P 50 : SOUND 129, T, 10, 15 : NEXT T : NEXT N : FOR T =
1 TO 3000 : GOTO 1470
480 RETURN

```

Cursor control

```

490 :
520 IF INKEY(67) = 32 THEN GOTO 1470
530 IF INKEY(62) = -1 THEN FLAG = 0 : PAPER 10 ELSE FLAG =
1 : RETURN
540 PEN CONTENTS(A, B)
550 IF CONTENTS(A, B) > 0 THEN BLOCK$ = PIECE$ ELSE BLOCK$
= SP$
560 GOSUB 930
570 PAPER COL(A, B)
580 GOSUB 930
590 IF INKEY(47) > -1 THEN RETURN
600 A = A + (INKEY(39) > -1) - (INKEY(31) > -1) : B = B + (
INKEY(69) > -1) - (INKEY(71) > -1)
610 IF A > 8 THEN A = 1
620 IF A < 1 THEN A = 8
630 IF B > 8 THEN B = 1
640 IF B < 1 THEN B = 8
650 GOTO 520

```

Replay game

```

660 :
690 FROMX = MX(COUNT) : FROMY = MY(COUNT)
700 COUNT = COUNT + 1
710 A = MX(COUNT)
720 B = MY(COUNT)
730 COUNT = COUNT + 1
740 IF MX(COUNT) = 0 THEN GOTO 1470
750 IF INKEY(67) > -1 THEN REPLAY = 0
760 RETURN

```

Draw Board

```
770 :
800 FOR T = 1 TO 8
810 FOR N = 1 TO 8
820 PAPER COL(N, T)
830 IF CONTENTS(N, T) = 0 THEN BLOCK$ = SP$
840 A = N : B = T
850 IF CONTENTS(N, T) > 0 THEN BLOCK$ = PIECE$
860 PEN CONTENTS(N, T)
870 GOSUB 930
880 NEXT N : NEXT T
890 RETURN
```

Block routine

```
900 :
930 LOCATE X(A, B), Y(A, B)
940 PRINT LEFT$(BLOCK$, 2)
950 LOCATE X(A, B), Y(A, B) + 1
960 PRINT RIGHT$(BLOCK$, 2)
970 RETURN
```

Initial initialisation

```
980 :
1010 MODE 0
1020 PEN 1 : INK 1, 18
1030 PAPER 0
1040 BORDER 1
1050 INK 0, 1 : CLS
1060 INK 2, 26
1070 INK 3, 0
1080 INK 4, 6
1090 REPLAY = 0
1100 INK 5, 20
1110 INK 7, 8
1120 DIM X(8, 8), Y(8, 8), COL(8, 8), CONTENTS(8, 8), MX(300
0), MY(3000)
```

Replay initialisation

```
1130 :
1160 CLS : COUNT = 1 : BPIECES = 0 : RPIECES = 0
1170 FOR T = 1 TO 8 : FOR N = 1 TO 8 : CONTENTS(N, T) = 0 :
NEXT N : NEXT T
1180 FOR N = 1 TO 8 STEP 2
1190 CONTENTS(N, 1) = 4 : CONTENTS(N + 1, 2) = 4 : CONTENTS(
N, 3) = 4
1200 CONTENTS(N + 1, 8) = 5 : CONTENTS(N, 7) = 5 : CONTENTS(
N + 1, 6) = 5
1210 NEXT N
1220 SYMBOL 240, 0, 3, 15, 31, 63, 63, 127, 127
```

```

1230 SYMBOL 241, 0 , 192, 240, 248, 252, 252, 254, 254
1240 SYMBOL 242, 127, 127, 63, 63, 31, 15, 3, 0
1250 SYMBOL 243, 254, 254, 252, 252, 248, 240, 192, 0
1260 PIECE$ = CHR$(240) + CHR$(241) + CHR$(242) + CHR$(243)
1270 SP$ = " ^^^^ "
1280 FOR T = 1 TO 8 STEP 2 : FOR N = 1 TO 8 STEP 2
1290 COL(N, T) = 2 : COL(N+ 1, T) = 3
1300 NEXT N : NEXT T
1310 FOR T = 2 TO 8 STEP 2 : FOR N = 1 TO 8 STEP 2
1320 COL(N, T) = 3 : COL(N + 1, T) = 2
1330 NEXT N : NEXT T
1340 FOR T = 1 TO 8
1350 FOR N = 1 TO 8
1360 X(N, T) = N * 2 : Y(N, T) = T * 2
1370 NEXT N : NEXT T
1380 GOSUB 800
1390 LOCATE 1, 1 : PAPER 0 : PEN 1 : PRINT STRING$(18, CHR$(
143))
1400 LOCATE 1, 18 : PAPER 0 : PEN 1 : PRINT STRING$(18, CHR$(
143))
1410 FOR T = 2 TO 17 : LOCATE 1, T : PRINT CHR$(143) : LOCAT
E 18, T : PRINT CHR$(143) : NEXT T
1420 LOCATE 6, 1 : PEN 7 : PRINT "DRAUGHTS"
1430 PEN 1 : RETURN

```

Menu

```

1440 :
1470 LOCATE 1, 25
1480 PRINT "REPLAY ^[Y/N]"
1490 A$ = INKEY$ : IF A$ = "" THEN GOTO 1490
1500 IF A$ = "Y" OR A$ = "y" THEN REPLAY = 1 : GOSUB 1160 :
GOTO 60
1510 IF A$ = "N" OR A$ = "n" THEN RUN
1520 GOTO 1490

```

ChexSum Tables

40	=	466	530	=	3417	1060	=	305
50	=	922	540	=	1520	1070	=	269
60	=	788	550	=	4510	1080	=	276
70	=	0	560	=	386	1090	=	855
100	=	341	570	=	1119	1100	=	302
110	=	322	580	=	386	1110	=	281
120	=	465	590	=	1609	1120	=	3877
130	=	308	600	=	6293	1130	=	0
140	=	1700	610	=	1418	1160	=	2747
150	=	2752	620	=	1421	1170	=	4425
160	=	1511	630	=	1420	1180	=	1295
170	=	4766	640	=	1423	1190	=	4401
180	=	2751	650	=	232	1200	=	4678
190	=	1511	660	=	0	1210	=	427
200	=	8340	690	=	3397	1220	=	1559
210	=	1791	700	=	1566	1230	=	2802
220	=	8240	710	=	1366	1240	=	1497
230	=	8200	720	=	1368	1250	=	2740
240	=	4906	730	=	1566	1260	=	3882
250	=	6719	740	=	2083	1270	=	729
260	=	11104	750	=	2283	1280	=	2597
270	=	12690	760	=	201	1290	=	2635
280	=	8336	770	=	0	1300	=	861
290	=	9375	800	=	991	1310	=	2598
300	=	466	810	=	985	1320	=	2635
310	=	308	820	=	1150	1330	=	861
320	=	1777	830	=	3104	1340	=	991
330	=	2485	840	=	1336	1350	=	1017
340	=	1436	850	=	3298	1360	=	3044
350	=	4691	860	=	1551	1370	=	861
360	=	2751	870	=	386	1380	=	256
370	=	1436	880	=	861	1390	=	2020
380	=	8265	890	=	201	1400	=	2048
390	=	1791	900	=	0	1410	=	3908
400	=	8164	930	=	1780	1420	=	1423
410	=	8124	940	=	1230	1430	=	436
420	=	7184	950	=	2039	1440	=	0
430	=	11104	960	=	1234	1470	=	310
440	=	6719	970	=	201	1480	=	1182
450	=	12916	980	=	0	1490	=	2190
460	=	8336	1010	=	219	1500	=	3525
470	=	9435	1020	=	531	1510	=	2204
480	=	201	1030	=	232	1520	=	437
490	=	0	1040	=	177			
520	=	1611	1050	=	406	TOTAL	=	333786